Green Lagrange Strain Tensor

We calculate the strain tensor if a unit cell with initial lattice parameter a_1, b_1, c_1, α_1, β_1, and γ is changed to a_2, b_2, c_2, α_2, β_2, and $\gamma + \Delta\gamma$.
For simplicity and 2D materials we take α_1= 90°, β_1=90°, and $c_1 = c_2$.

In[1]:=
```
(*Initial cell array*)
C0 = {{a1, 0, 0}, {b1 Cos[γ], b1 Sin[γ], 0}, {0, 0, c1}};
C0 = Transpose[C0];
MatrixForm[C0]
```

Out[3]//MatrixForm=

$$\begin{pmatrix} a_1 & \text{Cos}[\gamma]\, b_1 & 0 \\ 0 & \text{Sin}[\gamma]\, b_1 & 0 \\ 0 & 0 & c_1 \end{pmatrix}$$

In[4]:=
```
(*Final cell array*)
C1 = {{a2, 0, 0}, {b2 Cos[γ + Δγ], b2 Sin[γ + Δγ], 0}, {0, 0, c2}};
C1 = Transpose[C1];
MatrixForm[C1]
```

Out[6]//MatrixForm=

$$\begin{pmatrix} a_2 & \text{Cos}[\gamma+\Delta\gamma]\, b_2 & 0 \\ 0 & \text{Sin}[\gamma+\Delta\gamma]\, b_2 & 0 \\ 0 & 0 & c_2 \end{pmatrix}$$

In[7]:=
```
(*Deformation Gradient*)
```

In[8]:=
```
F = C1.Inverse[C0];
F // MatrixForm
```

Out[9]//MatrixForm=

$$\begin{pmatrix} \frac{a_2}{a_1} & -\frac{\text{Cot}[\gamma]\, a_2}{a_1} + \frac{\text{Cos}[\gamma+\Delta\gamma]\,\text{Csc}[\gamma]\, b_2}{b_1} & 0 \\ 0 & \frac{\text{Csc}[\gamma]\,\text{Sin}[\gamma+\Delta\gamma]\, b_2}{b_1} & 0 \\ 0 & 0 & \frac{c_2}{c_1} \end{pmatrix}$$

In[10]:=
```
(*Green Lagrange Strain Tensor*)
GLDouble = (Transpose[F].F - IdentityMatrix[3]);
DisplayForm[1/2 MatrixForm[Simplify[GLDouble]]]
```

Out[11]//DisplayForm=

$$\frac{1}{2}\begin{pmatrix} -1+\frac{a_2^2}{a_1^2} & \frac{\text{Csc}[\gamma]\, a_2\,(-\text{Cos}[\gamma]\, a_2 b_1+\text{Cos}[\gamma+\Delta\gamma]\, a_1 b_2)}{a_1^2 b_1} & 0 \\ \frac{\text{Csc}[\gamma]\, a_2\,(-\text{Cos}[\gamma]\, a_2 b_1+\text{Cos}[\gamma+\Delta\gamma]\, a_1 b_2)}{a_1^2 b_1} & -1+\frac{\text{Csc}[\gamma]^2\,\text{Sin}[\gamma+\Delta\gamma]^2\, b_2^2}{b_1^2}+\left(\frac{\text{Cot}[\gamma]\, a_2}{a_1}-\frac{\text{Cos}[\gamma+\Delta\gamma]\,\text{Csc}[\gamma]\, b_2}{b_1}\right)^2 & 0 \\ 0 & 0 & -1+\frac{c_2^2}{c_1^2} \end{pmatrix}$$

In[12]:=
```
(*For Simplicity let use also take γ=π/2*)
DisplayForm[1/2 MatrixForm[Simplify[GLDouble /. {γ → π/2}]]]
```

Out[12]//DisplayForm=

$$\frac{1}{2}\begin{pmatrix} -1+\frac{a_2^2}{a_1^2} & -\frac{\mathrm{Sin}[\Delta\gamma]\, a_2\, b_2}{a_1\, b_1} & 0 \\ -\frac{\mathrm{Sin}[\Delta\gamma]\, a_2\, b_2}{a_1\, b_1} & -1+\frac{b_2^2}{b_1^2} & 0 \\ 0 & 0 & -1+\frac{c_2^2}{c_1^2} \end{pmatrix}$$

In[13]:=
```
(*For hexagonal cell let use also take γ=π/3*)
DisplayForm[1/2 MatrixForm[Simplify[GLDouble /. {a2 → a1, γ → π / 3, Δγ → 0, c2 → c1}]]]
```

Out[13]//DisplayForm=

$$\frac{1}{2}\begin{pmatrix} 0 & \frac{-b_1+b_2}{\sqrt{3}\, b_1} & 0 \\ \frac{-b_1+b_2}{\sqrt{3}\, b_1} & -\frac{2\left(b_1^2+b_1\, b_2-2\, b_2^2\right)}{3\, b_1^2} & 0 \\ 0 & 0 & 0 \end{pmatrix}$$